
Introduction
A few months ago, I received an email from Joshua in Dallas, Texas, who wrote that "I have been reading that Oracle is releasing the eleventh version of its database technology. Very exciting! But here's the problem: I don't think I'll be able to use it for another two years. So why should I even care about the new (for me) PL/SQL features of this future release?"*

* Disclosure: my original answer to this question was published in Oracle Magazine in 2007. I have expanded on it for this paper.

This was, I thought, an excellent question. All too often, we technologists are attracted to technology and new features for the sake of having something new with which to play or about which to learn. And there sure isn't anything wrong with that. When it comes to PL/SQL developers, however, we tend to be a very pragmatic lot. You will hear questions like "How will a particular feature of Oracle help me satisfy a business requirement?" much more frequently than variations on "I wonder how I could convince my manager to try Extreme Programming's pair programming concept in our team?"

Joshua's question was a firm part of this tradition. We all know that eventually our companies will upgrade to Oracle11g (and then 12 and after that 13), but that process could take years. So we are generally quite content to pay the minimum attention necessary to futures and instead concentrate on the present.
Having said that, I do think that it makes an awful lot of sense to learn now about what Oracle Database 11g will have to offer you and your company in the future. The reason is very simple: once you see what is going to be available in Oracle Database 11g, you will probably want to change the way you write your code in Oracle9i and Oracle10g!

In this white paper, I will explore what I believe is the single-most important new PL/SQL feature in Oracle Database 11g is the PL/SQL result cache. Quite a mouthful, but then it is quite a feature. I believe that once you understand what this feature will be able to do for you, you will want to start getting ready now to quickly take advantage of this feature as soon as you upgrade to Oracle11g.
About the Result cache
Suppose I am on a team building a human resources application. The employees table is one of the key structures, holding all the data for all of our employees. Hundreds of users execute a large number of programs in the application that read from this table, and read from it very often. Yet the table changes relatively infrequently, perhaps once or twice an hour. As a result, the application code repeatedly retrieves from the block buffer cache what is for the most part static data, enduring the overhead of checking to see if the particular query has already been parsed, finding the data in the buffer, and returning it.

We need to improve the performance of querying data from the employees table.  Currently, we use the following function to return a row from the employees table:

FUNCTION one_employee (employee_id_in IN employees.employee_id%TYPE)
   RETURN employees%ROWTYPE
IS
   l_employee   employees%ROWTYPE;
BEGIN
   SELECT *

     INTO l_employee
     FROM employees
    WHERE employee_id = employee_id_in;
   RETURN l_employee;
EXCEPTION
   WHEN NO_DATA_FOUND
   THEN
      /* Return an empty record. */
      RETURN l_employee;
END one_employee;
On Oracle Database 11g, I can add a line to the header of this function as follows:

FUNCTION one_employee (employee_id_in IN employees.employee_id%TYPE)
   RETURN employees%ROWTYPE

   RESULT_CACHE RELIES_ON (employees)

This clause tells Oracle that it should remember (store in a special in-memory result cache) each record retrieved for a specific employee ID number. And when a session executes this function and passes in an employee ID that was previously stored, the PL/SQL runtime engine will not execute the function body, which includes that query.

Instead, it will simply retrieve the record from the cache and return that data immediately.

The result is much faster retrieval time. 

In addition, by specifying "RELIES_ON (employees)", we inform the Oracle that if any session commits changes to that table, any data in the result cache drawn from the table must be invalidated. The next call to the one_employee function would then have to execute the query and retrieve the data fresh from the table.

Finally, this cache is not per session, but is available to all sessions connected to the instance. 

Prior to Oracle Database 11g, I could accomplish a similar kind of caching using package-level collections, but this cache would be session-specific and located in the User Global Area. Which means that if I have 1000 different sessions running the application, I could use up an enormous amount of memory in addition to that consumed by the SGA. 

The PL/SQL result cache minimizes the amount of memory needed to cache and share this data across all sessions. This low memory profile, plus the automatic purge of cached results whenever changes are committed, makes this feature of Oracle Database 11ga very practical method for optimizing performance in our applications. 
Analyze performance and PGA memory impact

To test the improvement in performance and impact on PGA memory over repeated querying of the data, I put together a set of scripts that compares three different way to retrieve a row of employee data. you can obtain these files by visiting:
http://www.ToadWorld.com/SF (my PL/SQL Obsession page)

Then click on the "Trainings, Seminars and Presentations" link and then click on the "demo.zip" link. This zip archive contains approximately 2000 scripts that are referenced in my various trainings. You will find in this demo.zip archive the "11g_emplu.tst" script, which references all other files that must be created to reproduce what you will find below.

Here are my three different approaches, and the associated code:

1. Execute the query N times (no PL/SQL-based caching).
I will place my query inside a function, and call that function. I will, therefore, execute the query repeatedly. The RDBMS will, of course, optimize performance through caching of both the data and pre-parsed cursor.
PACKAGE BODY emplu1
IS
   FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)
      RETURN employees%ROWTYPE
   IS
      onerow_rec   employees%ROWTYPE;
   BEGIN
      SELECT *

        INTO onerow_rec
        FROM employees
       WHERE employee_id = employee_id_in;
      RETURN onerow_rec;
   END;
END;
/
2. Cache all the rows of the employees table in a packaged collection and the retrieve the data from that cache. 
In the second approach, I create a package-level collection, employee_cache, and fill it with all the data of my employees table as part of my package initialization. This collection holds that data while my session is active, even if no program is using it. I use the primary key of the employee table as the index value in the collection. As a result, my lookup function doesn't execute a query. Instead, it simply returns the element at the specified index value.

PACKAGE BODY emplu2
IS
   TYPE employee_tt IS TABLE OF employees%ROWTYPE
      INDEX BY PLS_INTEGER;
   employee_cache   employee_tt;
   FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)
      RETURN employees%ROWTYPE
   IS
   BEGIN
      RETURN employee_cache (employee_id_in);
   END onerow;
   PROCEDURE load_cache
   IS
   BEGIN
      FOR rec IN (SELECT   *

                      FROM employees)
      LOOP
         employee_cache (rec.employee_id) := rec;     

      END LOOP;
   END load_cache;
BEGIN
   load_cache;
END emplu2;
3. Use the PL/SQL Result Cache to avoid repetitive querying.

Finally, I take my original emplu1.onerow function, and transform it into a result cache function:

PACKAGE BODY emplu11g
IS
   FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)
      RETURN employees%ROWTYPE
      result_cache relies_on (employees)
   IS
      onerow_rec   employees%ROWTYPE;
   BEGIN
      SELECT *

        INTO onerow_rec
        FROM employees
       WHERE employee_id = employee_id_in;
      RETURN onerow_rec;
   END;
END emplu11g;
With these three packages, I calculate elapsed time using DBMS_UTLITY.GET_TIME (via the PLVtmr package) and I compute memory consumption using v$ views (via the my_session package). 

When I run the 11g_emplu.tst script, roughly 5 or 6 seconds elapse and then I see results like this:

PGA before tests are run:

session PGA:  910860

Execute query each time Elapsed: 4.5 seconds. Factored: .00005 seconds.

session PGA:  910860

Cache table in PGA memory Elapsed: .11 seconds. Factored: 0 seconds.

session PGA: 1041932

Oracle Database 11g result cache Elapsed: .27 seconds. Factored: 0 seconds.

session PGA: 1041932

Here are my conclusions from this admittedly incomplete analysis:

The Oracle Database 11g PL/SQL Result Cache is, indeed, much faster that repetitive querying. In this test, it was over an order of magnitude faster.

A packaged collection cache is even faster, most likely because the PL/SQL runtime engine can access the data from UGA and PGA memory, rather than SGA memory. Unfortunately this also means that the consumption of memory occurs on a per-session basis, which is not very scalable.

The packaged collection approach consumed additional UGA and PGA memory, but the Oracle Database 11g Result Cache did not.

And then, of course, there are the other key advantages of the Result Cache: automatic invalidation of cache contents when a dependent table is changed; the fact that the cache is shared across sessions; and the application of the least recently used algorithm to the memory in the cache.

So why should you care now?

"All right," you may be saying to yourself. "It's cool. Super cool. I still can't use it for two years or more, so what good does that do me now?"

You may not be able to use the PL/SQL result cache yet, but you can write your code now so that when you upgrade to Oracle Database 11g, you will be able to quickly and easily use this cache in your application code.

In other words, you can and should prepare now for this future feature.

How do you do that? By placing all your queries (at least those against tables that change infrequently, but are queried often) inside functions, so that you can easily add the RESULT_CACHE clause.

Think about it: today, you probably don't do that. Instead, whenever you need data from the database, you write the required query, right there in the application logic you are writing (whether that application logic resides in the backend – other PL/SQL programs – or the frontend, in languages like Java). 

And that same query (or some minor variation on it) will likely appear in multiple places in your application code. Why not? It is so easy to write those SQL statements; that's one of the beauties of PL/SQL.

But that ease of use in executing SQL inside PL/SQL leads to us all taking SQL for granted, and when you upgrade to Oracle Database 11g, you will pay the price. 

If after upgrading, you then want to take advantage of RESULT_CACHE, you will have to either find every affected SQL statement and either:

(a) Put the RESULT_CACHE hint inside that query (that's right, this feature is available natively within SQL, as well as for functions) or

(b) Construct the function and put the query inside it; find each of the applicable queries; replace the query with the function call.

Certainly either of those approaches are eminently doable, but it is also very unlikely to happen. IT managers are loathe to go into existing, working production code and "upset the apple cart" by making lots of changes. 

If, conversely, you start right now, in Oracle 9i or Oracle Database 10g, to place your queries inside functions, then when you upgrade to Oracle Database 11g, you will almost instantly be able to upgrade your code to use this fantastic new feature. 

And, best of all, the application code that calls the function will not have to be changed at all! Your manager will be very impressed.

And that is why you should learn about the new features of Oracle Database 11g. Today. Right now!

Taking Things a Step Further

So...you are excited about the result cache feature and you are from now on going to hide those queries inside functions, so that when Oracle11g arrives, you can quickly put this feature to use.
Well, I say: why wait till Oracle11g arrives? You can use another super-cool feature of PL/SQL, conditional compilation, which was added in Oracle 10g Release 2 (but is also available in recent patch releases of both Oracle 10g Release 1 and Oracle 9i Release 2), to implement the result cache but make sure it "hibernates" (is not included in your compiled code) until you are running Oracle11g.

Here's how it works:

To use the conditional compilation feature of PL/SQL, you will include "$" syntax in your code, including $IF, $THEN, $END, $ERROR and $$identifier, which in essence tell the PL/SQL compiler which code you want to actually be compiled.
Conditional compilation is a very powerful capability, but it is also somewhat complicated. A full explanation is well outside the scope of this paper. For full details and many use cases, check out PL/SQL Product Manager Bryn Llewellyn's white paper on this topic, available on OTN.

I will simply show you how you can use it to implement automatic upgrades to the result cache.
So I am currently on Oracle 10g Release 2. I have moved my query inside a function, but I cannot use package-based caching because the contents of the table do change throughout the day. Instead, I will get ready for the result cache! I change my package specification to include the result_cache keyword, but inside a conditional compilation clause:


PACKAGE emplu
IS
   FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)
      RETURN employees%ROWTYPE
   $IF DBMS_DB_VERSION.VERSION >= 11
   $THEN
      RESULT_CACHE
   $END
   ;
END emplu;
The $IF clause specifies that at the time of compilation, the compiler should check the value of the DBMS_DB_VERSION.VERSION constant. If the value is at least 11, then include the result_cache clause. Otherwise leave it out.

Every version of Oracle that supports conditional compilation also installs the DBMS_DB_VERSION package, which contains absolute and relative version information.

I do something similar in the package body:

PACKAGE BODY emplu
IS
   FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)
      RETURN employees%ROWTYPE
   $IF dbms_db_version.version >= 11
   $THEN
      RESULT_CACHE RELIES_ON (employees)
   $END
   IS
      onerow_rec   employees%ROWTYPE;
   BEGIN
      SELECT *

        INTO onerow_rec
        FROM employees
       WHERE employee_id = employee_id_in;
      RETURN onerow_rec;
   END;
END emplu;
So in my current project, when I compile this code on Oracle10g, there is no result_cache and life proceeds as before. Now suppose that I move on to another project, a few years go by and finally the database for our little HR system is going to be upgraded to Oracle11g.

When the emplu package is compiled on Oracle11g, conditional compilation determines that the result_cache clause should be included and voila! The software is automatically upgraded without any need for programmer intervention!

Now if that idea is a bit too scary for you to consider, then you might want to implement the package body like this: 
PACKAGE BODY emplu
IS
   FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)
      RETURN employees%ROWTYPE
   $IF dbms_db_version.version >= 11
   $THEN
      result_cache relies_on (employees)
   $END
   IS
      onerow_rec   employees%ROWTYPE;
   BEGIN
   $IF dbms_db_version.version >= 11
   $THEN
      RAISE_APPLICATION_ERROR (

         -20000,

         'Review this use of RESULT_CACHE!' ||

         ' And Call Steven if any questions...');   

   $END   

      SELECT *

        INTO onerow_rec
        FROM employees
       WHERE employee_id = employee_id_in;
      RETURN onerow_rec;
   END;
END emplu;
Now when anyone tries to run the emplu.onerow function, it will raise an exception and force you to explicit approve or rejct the use of the result cache.

Alternatively, you could use the $ERROR directive to force a compilation error on Oracle11g and force a decision regarding the use of the result cache:

SQL> CREATE OR REPLACE PACKAGE emplu

  2  IS

  3     FUNCTION onerow (employee_id_in IN employees.employee_id%TYPE)

  4        RETURN employees%ROWTYPE

  5     $IF DBMS_DB_VERSION.VERSION >= 11

  6     $THEN

  7        $ERROR 'This function is an excellent candidate for the result cache.' ||

  8               ' Decide on whether or not to use it here.' 

  9     $END

 10     ;

 11  END emplu;

 12  /

Warning: Package created with compilation errors.

SQL> sho err

Errors for PACKAGE EMPLU:

LINE/COL ERROR

-------- -----------------------------------------------------------------

7/7      PLS-00179: $ERROR: This function is an excellent candidate for

         the result cache. Decide on whether or not to use it here.

In Conclusion

Additional features from Oracle11g will be covered at Collaborate08, but the function result cache discussed above is the most important of these features. I hope that I have convinced you that even if you will not be able to actually use Oracle11g features for active development for years to come, that its features should have an impact on the way you write your code today.

